



PRODUCT NAME

TM58XX

TITLE

TM58XX RC 测温

APPLICATION NOTE

1. 程序
2. 应用线路图

```

;=====
;Title:TM58xx Application note
;file name----RC_MEASURE_T.asm
;
;Revision History:
;Rev:      Date:      Reason:
;1.0      4/21/2004   original
;
;
;=====
;
;晶振：4MHZ
;看门狗（WATCHDOG）：禁止（DISABLE）
;功能:利用 RC 充放电实现测温功能
;测试温度范围：0~40°C,测试温度由两位数码管显示,测量
;精度：+/-1°C.使用传感器：AT-103 型温度传感器。程序中
;每秒进行三次测温,每隔 1s 刷新一次温度显示,程序中设有传感
;器自检功能,当传感器开路或短路时数码管秒闪显示"E"表示
;出错.待更换传感器并重新,上电后系统可恢复正常。系统开机
;4s 内在传感器未出错的情况下,数码管显示"88",而不显示测
;温结果,以此表示数码管显示正常。程序中对测温结果的处理
;方式如下:
;将对传感器进行 RC 充放电所得的计数值(16 位,存放在(rtalh,
;rtall)中)和对比对电阻进行 RC 充放电所得的计数值(16 位,存
;放(rhalh,rhall)中)进行 16 位除法(即(rtalh,rtall)/
;(rhalh,rhall)),相除的结果保存在(divh,divl)中,其结果
;形式为 1 位整数和三位小数,其中 divh 高半字节存放整数位,低半
;字节存放小数十分位,divl 高半字节存放小数百分位,低半字节
;存放小数千分位.温度表格 temprature_h 中存放 divh 值,
;temprature_l 中存放 divl 值。将 RC 充放电所测得的计数值相除
;后的结果与温度表格中的值相比较即可得到所测的温度值。
;
;
;=====
;      include "d:\58pc10.inc"    ;引入包含文件
;
;
;=====
;位置定义

```

```

;=====
ctr_flag      equ      08h   ;显示,测温控制标志
time_temp     equ      09h
time_buf      equ      0ah
sec_acc       equ      0bh   ;秒计数器
;-----
dis_h         equ      0ch   ;温度高位显示段码暂存
dis_l         equ      0dh   ;温度低位显示段码暂存
;-----
porta_buf     equ      0eh   ;porta 口缓冲器
oc_value      equ      0fh   ;测试温度值暂存
rhall         equ      10h   ;测试比对电阻计数值暂存
rhalh         equ      11h   ;测试比对电阻计数值暂存
rtall         equ      12h   ;测试温度传感器计数值暂存
rtalh         equ      13h   ;测试温度传感器计数值暂存
divl          equ      14h   ;测试计数值比较结果暂存
divh          equ      15h   ;测试计数值比较结果暂存
buffer1       equ      16h
buffer0       equ      17h
dly           equ      18h   ;测试比对电阻延时 0.196608s 寄存器
time_cnt      equ      19h   ;开机 4s 不显示测温数据时间计数器
;-----
byte          equ      1ah
r0            equ      1bh
r1            equ      1ch
temp          equ      1dh
count         equ      1eh
;-----
min           equ      byte
mid           equ      r0
max           equ      r1
buf           equ      rtalh
i1           equ      count
i2           equ      time_cnt
;
;
;
;=====
;porta 口    I/O 定义
;=====

```

```

led_com2      equ      00h
led_com1      equ      01h
sensor        equ      02h
calibration   equ      03h
;
;
;=====
;ctr_flag 位定义
;=====
display       equ      00h ;数码管高低位显示控制标志,"0":显示高位,"1":显示低位
oc_crt        equ      01h ;测温控制标志
rt_ct_crt     equ      02h ;测试传感器和比对电阻控制标志,"0":传感器,"1":比对电阻
error_crt     equ      03h ;传感器出错控制标志
error_show    equ      04h ;出错信息"E"显示控制标志
ct_delay      equ      05h ;测试比对电阻延时控制标志
data_ctr      equ      06h ;测温数据处理控制标志
;
;
;=====
;=====
;prog. start here
;=====
;=====
org           3ffh
lgoto        main
org           000h
;=====
;rt/tc 的高位表,注释中 R 为传感器对应的阻值
;=====
temperature_h:
    addampc,1
    retla    56    ;"0'c"----R:27.28k
    retla    53    ;"1'c"----R:26.13K
    retla    51    ;"2'c"----R:25.03K
    retla    49    ;"3'c"----R:24.00K
    retla    47    ;"4'c"----R:23.00K
    retla    45    ;"5'c"----R:22.05K
    retla    43    ;"6'c"----R:21.15K
    retla    41    ;"7'c"----R:20.30K

```

retla	40	;"8'c"----R:19.48K
retla	38	;"9'c"----R:18.70K
retla	37	;"10'c"----R:17.96K
retla	35	;"11'c"----R:17.24K
retla	34	;"12'c"----R:16.65K
retla	32	;"13'c"----R:15.90K
retla	31	;"14'c"----R:15.28K
retla	30	;"15'c"----R:14.69K
retla	29	;"16'c"----R:14.12K
retla	28	;"17'c"----R:13.58K
retla	27	;"18'c"----R:13.06K
retla	26	;"19'c"----R:12.56K
retla	25	;"20'c"----R:12.09K
retla	24	;"21'c"----R:11.63K
retla	23	;"22'c"----R:11.20K
retla	22	;"23'c"----R:10.78K
retla	21	;"24'c"----R:10.38K
retla	20	;"25'c"----R:10.00K
retla	19	;"26'c"----R:9.632K
retla	19	;"27'c"----R:9.281K
retla	18	;"28'c"----R:8.944K
retla	17	;"29'c"----R:8.622K
retla	17	;"30'c"----R:8.313K
retla	16	;"31'c"----R:8.014K
retla	16	;"32'c"----R:7.728K
retla	15	;"33'c"----R:7.454K
retla	14	;"34'c"----R:7.192K
retla	14	;"35'c"----R:6.940K
retla	13	;"36'c"----R:6.699K
retla	13	;"37'c"----R:6.467K
retla	12	;"38'c"----R:6.245K
retla	12	;"39'c"----R:6.032K
retla	12	;"40'c"----R:5.827K

;
;
;

=====

;rt/tc 的低位表

=====

temprature_l:

addam	pc,1	
retla	15	;"0'c"
retla	74	;"1'c"
retla	61	;"2'c"
retla	48	;"3'c"
retla	15	;"4'c"
retla	32	;"5'c"
retla	68	;"6'c"
retla	79	;"7'c"
retla	02	;"8'c"
retla	50	;"9'c"
retla	03	;"10'c"
retla	68	;"11'c"
retla	27	;"12'c"
retla	84	;"13'c"
retla	61	;"14'c"
retla	32	;"15'c"
retla	19	;"16'c"
retla	17	;"17'c"
retla	12	;"18'c"
retla	10	;"19'c"
retla	02	;"20'c"
retla	04	;"21'c"
retla	20	;"22'c"
retla	30	;"23'c"
retla	40	;"24'c"
retla	68	;"25'c"
retla	93	;"26'c"
retla	34	;"27'c"
retla	68	;"28'c"
retla	78	;"29'c"
retla	42	;"30'c"
retla	85	;"31'c"
retla	01	;"32'c"
retla	44	;"33'c"
retla	93	;"34'c"

```

retla    40      ;"35'c"
retla    95      ;"36'c"
retla    50      ;"37'c"
retla    96      ;"38'c"
retla    45      ;"39'c"
retla    05      ;"40'c"

```

```

;
;
;

```

```

;=====

```

```

;DIGI_LED 显示段码表

```

```

;=====

```

```

dis_table:

```

```

    addam    pc
retla    40h      ;"0"
retla    79h      ;"1"
retla    24h      ;"2"
retla    30h      ;"3"
retla    19h      ;"4"
retla    12h      ;"5"
retla    02h      ;"6"
retla    78h      ;"7"
retla    00h      ;"8"
retla    10h      ;"9"

```

```

;
;
;

```

```

;=====

```

```

;8 位二进制数转 BCD 码子程序

```

```

;=====

```

```

btobcd8bit:

```

```

    bcm      status,c
    movla    08h      ;循环移位次数 8 次
    movam    count
    clrm     r0
    clrm     r1

```

```

loop8:

```

```

    rlm      byte
    rlm      r1
    rlm      r0      ;byte 左移一位至 r1,r0 中
    decmsz   count
    lgoto    adjdec  ;将 r0,r1 移位结果进行 bcd 调整

```

```

    ret
adjdec:
    movla    r1
    movam    fsr
    lcall    adjbcd
    movla    r0
    movam    fsr
    lcall    adjbcd
    lgoto    loop8
;-----
adjbcd:
    movla    03h
    addam    indf,w           ;低半字节+3>7 吗?
    Movam    temp
    btmsc    temp,3
    movam    indf           ;是,低半字节+3>7,lcd+3--->lcd
    movla    30h
    addam    indf,w           ;高半字节+3>7 吗?
    Movam    temp
    Btmsc    temp,7
    movam    indf           ;是,高半字节+3>7,msd+3--->msd
    ret
;
;
;=====
;充电时间测量子程序
;=====
test_oc:
    clrm     buffer0
    clrm     buffer1
    btmsc    ctr_flag,rt_ct_crt ;判断处于测试传感器还是比对电阻过程?
    lgoto    sensor_res
;-----
calibration_res:           ;测试比对电阻充放电过程
    movla    04h
    iodir    porta         ;设 porta,sensor 口为输入,porta_calibration 口为输出
    bsm      porta_buf,calibration
    movm     porta_buf,w
    movam    porta

```

```

ct_loop:
    incmsz    buffer1
    lgoto    ct_endchk
    incmsz    buffer0
    lgoto    ct_endchk
    lgoto    have_error
ct_endchk:
    btmss    porta,sensor
    lgoto    ct_loop
    bcm      ctr_flag,rt_ct_crt
    bcm      ctr_flag,oc_crt
    lgoto    end_m
;-----
sensor_res:                                ;测试传感器充放电过程
    movla    08h
    iodir    porta                        ;设 porta,sensor 口为输出,porta_calibration 口为输入
    bsm      porta_buf,sensor
    movm     porta_buf,w
    movam    porta
rt_loop:
    incmsz    buffer1
    lgoto    rt_endchk
    incmsz    buffer0
    lgoto    rt_endchk
    lgoto    have_error
rt_endchk:
    btmss    porta,calibration
    lgoto    rt_loop
    bsm      ctr_flag,rt_ct_crt
;-----
end_m:
    bcm      porta_buf,sensor
    bcm      porta_buf,calibration
    movm     porta_buf,w
    movam    porta
    clra
    iodir    porta
    movm     buffer0,w

```

```

    btmss    status,z
    ret
    movla   06h
    subam   buffer1,w
    btmsc   status,c           ;buffer1<=6,测温电路短路?
    ret
;-----
have_error:           ;测温电路出错自诊断过程
    bcm    porta_buf,sensor
    bcm    porta_buf,calibration
    movm   porta_buf,w
    movam  porta
    clra
    iodir  porta
    bsm    ctr_flag,error_crt ;传感器开路,置传感器出错控制标志
movla 0ffh
    movam  dis_h
    movla  06h
    movam  dis_l
    clrm   sec_acc
    bsm    ctr_flag,error_show
    ret
;
;
;
;=====
;主程序
;=====
main:
    clrm   time_cnt
    clrm   dly
    clrm   sec_acc
    clrm   ctr_flag
    bsm    ctr_flag,oc_crt
    clrm   dis_l           ;开机数码管显示"00"
    clrm   dis_h
    movam  portb           ;portb 口输出"0ffh"
    movla  03h
    movam  porta           ;porta 口输出"03h"
    movam  porta_buf

```

```

        clra                ;清累加器
        iodir    portb     ;portb 口设为输出口
        iodir    porta     ;porta 口设为输出口
        movla    05h
        select                ;预分频器分配给 tmr0,分频比设定为 1:64
        movm     tmr0,w
        movam    time_temp   ;读取 tmr0 值送 time_temp 中
;-----
;-----
read_time:
        movm     tmr0,w     ;读取 tmr0 的值,结果放入累加器中
        xoram    time_temp,w ;tmr0 的值与 time_temp 值相异或,结果放入累加器中
        movam    time_buf   ;将 tmr0 与 time_temp 异或的结果送入 time_buf 中
        xoram    time_temp,f ;tmr0 的值送入 time_temp 中
        btms    time_buf,7  ;判断 128*64us/1000=8.192ms 是否已到
        lgoto    read_time  ;8.192ms 未到 , 则继续判断时间
;-----
display_oc:                ;8.192ms 已到,进入显示温度程序段
        btms    ctr_flag,display ;判断数码管应显示高位还是低位?
        lgoto    show_h
;-----
show_l:                    ;数码管显示低位
        bcm     ctr_flag,display ;清数码管高低位显示控制标志,使其指向高位
        bsm     porta_buf,led_com2;关高位数码管
        movm    porta_buf,w
        movam   porta
        movm    dis_l,w
        movam   portb                ;低位显示段码送 portb 口
        bcm     porta_buf,led_com1;开低位数码管
        movm    porta_buf,w
        movam   porta
        lgoto    deal_time          ;转定时时间处理程序段
;-----
show_h:                    ;数码管显示高位
        bsm     ctr_flag,display ;清数码管高低位显示控制标志,使其指向低位
        bsm     porta_buf,led_com1;关低位数码管
        movm    porta_buf,w
        movam   porta
        movm    dis_h,w

```

```

    movam    portb                ;高位显示段码送 portb 口
    bcm     porta_buf,led_com2;开高位数码管
    movm    porta_buf,w
    movam    porta
;-----
deal_time:                ;定时时间处理程序段
    incm    sec_acc
    btmss   ctr_flag,error_crt ;判断传感器是否出错
    lgoto   sensor_no_error ;转 1s 计时程序段
;-----
;-----
sensor_error:            ;传感器出错,进行 0.5s 计时处理
    movla   3dh                ;61*8.192ms=499.712ms 到否?
    xoram   sec_acc,w
    btmss   status,z
    lgoto   read_time          ;0.5s 时间未到,继续判断时间
    clrm    sec_acc            ;0.5s 时间到,清 sec_acc
    btmss   ctr_flag,error_show;判断是否正在显示出错信息"E"?
    lgoto   set_e              ;"E"正处于不显示状态,转设置"E"显示程序段
;-----
clear_e:                  ;"E"正处于显示状态,设置"E"不显示
    bcm     ctr_flag,error_show;清出错信息"E"显示控制标志使其指向不显示状态
    movla   0ffh
    movam   dis_l              ;"0ffh"段码送 dis_l,使"E"不显示
    lgoto   read_time          ;继续判断时间
;-----
set_e:                    ;"E"正处于不显示状态,设置"E"显示
    bsm     ctr_flag,error_show;置出错信息"E"显示控制标志使其指向显示状态
    movla   06h
    movam   dis_l              ;"06h"段码送 dis_l,使"E"显示
    lgoto   read_time          ;继续判断时间
;-----
;-----
sensor_no_error:         ;传感器未出错,进行 0.32768s 计时处理
    movla   28h                ;40*8.192ms/1000s=0.32768s 到否?
    xoram   sec_acc,w
    btmss   status,z
    lgoto   $+.12
    bsm     ctr_flag,oc_crt    ;置测温控制标志,使测温有效

```

```

clrm    sec_acc      ;0.32768s 时间到,清 sec_acc
btmsc   ctr_flag,data_ctr ;判断测温数据处理控制标志是否无效
lgoto   $+8         ;有效
incm    time_cnt    ;无效
movla   0ch         ;判断 12*0.32768s=3.93216s 测温数据不处理时间是否已到?
xoram   time_cnt,w
btmss   status,z
lgoto   $+3         ;3.93216s 未到
clrm    time_cnt    ;3.93216s 已到
bsm     ctr_flag,data_ctr
btmss   ctr_flag,ct_delay ;判断是否处于 0.196608s 延迟时间段?
lgoto   test_tempratuer ;不处于 0.196608s 延迟时间段 , 转测温程序段
incm    dly         ;处于 0.196608s 延迟时间段
movla   14h         ;判断 20*8.192ms/1000=0.196608s 延迟时间到否?
xoram   dly,w
btmss   status,z
lgoto   test_tempratuer ;0.196608s 延迟未到,转测温程序段
clrm    dly         ;0.196608s 延迟已到
bcm     ctr_flag,ct_delay
;-----
;-----
test_tempratuer:          ;测温程序段
    btmss   ctr_flag,oc_crt ;判断测温是否有效?
    lgoto   read_time      ;测温无效,继续判断时间
    btmsc   ctr_flag,rt_ct_crt ;测温有效,判断测传感器还是比对电阻?
    lgoto   test_ct        ;测比对电阻有效,转测试比对电阻程序段
;-----
test_rt:                  ;测传感器有效,处理测传感器程序
    lcall   test_oc
    btmsc   ctr_flag,error_crt
    lgoto   read_time
    bsm     ctr_flag,ct_delay ;置延迟 0.196608s 控制标志
    movm    buffer0,w
    movam   rtalh
    movm    buffer1,w
    movam   rtall
    lgoto   read_time
;-----

```

```

test_ct:                                ;测比对电阻有效,处理测比对电阻程序
    btmsc    ctr_flag,ct_delay;判断延迟 0.196608s 是否已到?
    lgoto    read_time    ;时间未到,继续判断时间
    lcall    test_oc
    btmsc    ctr_flag,error_crt
    lgoto    read_time
    movm    buffer0,w
    movam    rhalh
    movm    buffer1,w
    movam    rhall
    btmsc    ctr_flag,data_ctr ;判断测温数据处理控制标志是否有效?
    lgoto    read_time    ;标志无效,继续判断时间
;-----
div:                                       ;测温数据处理程序段,(rtalh,rtall)/(rhalh,rhall)
                                           ;相除后结果放在 divh 和 divl 中,其中 divh 高 4 位放结果的
                                           ;整数位,低 4 位放结果小数部分的十分位,divl 高 4 位放结果
                                           ;小数部分的百分位,低 4 位放结果小数部分的千分位
    clrm    divl
    clrm    divh
;-----
div1:
    lcall    sub
    btmsc    status,c
    goto     div2    ;(rtalh,rtall)减(rhalh,rhall)不够减
    movla    10h
    addam    divh,f    ;(rtalh,rtall)减(rhalh,rhall)够减,divh 高半字节加 1
    goto     div1
;-----
div2:
    movm    buffer1,w    ;(rtalh,rtall)减(rhalh,rhall)不够减,恢复 rtall 和 rtalh 相减前的值
    movam    rtall
    movm    buffer0,w
    movam    rtalh
    lcall    multi_10    ;(rtalh,rtall)乘 10
;-----
div3:
    lcall    sub
    btmsc    status,c

```

```

    goto    div4          ;(rtalh,rtall)减(rhalh,rhall)不够减
    movla   01h
    addam   divh,f        ;(rtalh,rtall)减(rhalh,rhall)够减,divh 低半字节加 1
    lgoto   div3
; -----
div4:
    movm    buffer1,w     ;(rtalh,rtall)减(rhalh,rhall)不够减,恢复 rtall 和 rtalh 相减前的值
    movam   rtall
    movm    buffer0,w
    movam   rtalh
    lcall   multi_10      ;(rtalh,rtall)乘 10
; -----
div5:
    lcall   sub
    btmss   status,c
    goto    div6          ;(rtalh,rtall)减(rhalh,rhall)不够减
    movla   10h
    addam   divl,f        ;(rtalh,rtall)减(rhalh,rhall)够减,divl 高半字节加 1
    lgoto   div5
; -----
div6:
    movm    buffer1,w     ;(rtalh,rtall)减(rhalh,rhall)不够减,恢复 rtall 和 rtalh 相减前的值
    movam   rtall
    movm    buffer0,w
    movam   rtalh
    lcall   multi_10      ;(rtalh,rtall)乘 10
; -----
div7:
    lcall   sub
    btmss   status,c
    goto    get_oc        ;(rtalh,rtall)减(rhalh,rhall)不够减
    movla   01h
    addam   divl,f        ;(rtalh,rtall)减(rhalh,rhall)够减,divl 低半字节加 1
    lgoto   div7
; -----
get_oc:
    clrm    oc_value
loopz1:
    movm    oc_value,w

```

```
lcall    temprature_h
movam    rhalh
movm     oc_value,w
lcall    temprature_l
movam    rhall
movm     divh,w
movam    rtalh
movm     divl,w
movam    rtall
lcall    sub
btmsc    status,c
lgoto    get_mid_value
loopz2:
incm     oc_value
movla    28h          ;40'C
subam    oc_value,w
btmss    status,c
lgoto    loopz1      ;oc_value<40'C
movla    28h
movam    oc_value    ;oc_value>=40'C,则 28h 付给 oc_value
; -----
get_mid_value:
movla    00h
xoram    i2,w
btmss    status,z
lgoto    mid_max
movm     oc_value,w
movam    min
incm     i2
lgoto    read_time
mid_max:
movla    01h
xoram    i2,w
btmss    status,z
lgoto    max_max
movm     oc_value,w
movam    mid
```

```

        incm    i2
        lgoto   read_time
max_max:
        movm    oc_value,w
        movam   max
        clrm    i2
        lcall   sort_sub    ;将三次测温结果按由小到大的顺序排列
        movm    mid,w      ;取三次测温结果的中间值
        movam   oc_value
; -----
get_oc_code:    ;测温结果的显示段码处理程序段
        movm    oc_value,w
        movam   byte
        lcall   btobcd8bit
        movm    r1,w
        andla   0fh
        lcall   dis_table
        movam   dis_l
        movm    r1,w
        andla   0f0h
        btmss   status,z
        lgoto   get_hbit_code
        movla   0ffh
        movam   dis_h
        lgoto   read_time
get_hbit_code:
        movam   byte
        swapm   byte,w
        lcall   dis_table
        movam   dis_h
        lgoto   read_time
;
;
;
;-----
        org    200h
;
;
;
;-----
;16 位二进制数加法子程序

```

;该子程序实现将 buffer0(高 8 位),buffer1(低 8 位)表示的 16 位二进制数
;和 rtalh(高 8 位),rtall(低 8 位)表示的 16 位二进制数相加,最终结果存入
;rtalh(高 8 位),rtall(低 8 位)中.

;(rtalh,rtall)=(rtalh,rtall)+(buffer0,buffer1)

;该程序要求最终结果不可超过 16 位二进制数所能表示的最大值(65535),否
;则结果出错

;=====

add:

movm buffer1,w

addam rtall

btmsc status,c

incm rtalh

movm buffer0,w

addam rtalh

ret

;
;
;

;=====

;16 位无符号二进制数减法子程序

;该程序实现 rtalh(高 8 位),rtall(低 8 位)表示的 16 位二进制数和

;rhalh(高 8 位),rhall(低 8 位)表示的 16 位二进制数相减,减后的结

;果放入 rtalh(高 8 位)和 rtall(低 8 位)中.

;(rtalh,rtall)=(rtalh,rtall)-(rhalh,rhall)

;因为无符号二进制数相减,(rtalh,rtall)表示的二进制数应大于

;(rhalh,rhall)表示的二进制数,否则结果出错

;=====

sub:

movm rtalh,w

movam buffer0

movm rtall,w

movam buffer1

;-----

;低 8 位相减

movm rhall,w

subam rtall,f

btmss status,c ;判断 rtall 减 rhall 是否够减?

lgoto have_carry ;rtall 不够减,向 rhall 有借位,转去执行 rtalh 减 1

decrease: ;高 8 位相减

movm rhalh,w

```

        subam    rtalh,f
        ret
have_carry:                ;高 8 位减 1
        movla    01h
        subam    rtalh,f
        btmsc    status,c    ;判断 rtalh 减 1 是否够减?
        lgoto    decrease    ;rtalh 减 1 够减
        ret                ;rtalh 减 1 不够减,结果出错,不再继续减
;
;
;=====
;16 位二进制数乘 10 子程序
;该子程序实现将 rtalh(高 8 位)和 rtall(低 8 位)表示的 16 位二进制数乘 10
;(rtalh,rtall)<=(((rtalh,rtall)*2)*2)*2+(rtalh,rtall)+(rtalh,rtall)
;=====
multi_10:
        movm     rtall,w
        movam    buffer1
        movm     rtalh,w
        movam    buffer0
        ;-----
        bcm      status,c    ;(rtalh,rtall)*2
        rlm      rtall
        rlm      rtalh
        ;-----
        bcm      status,c    ;(rtalh,rtall)*2
        rlm      rtall
        rlm      rtalh
        ;-----
        bcm      status,c    ;(rtalh,rtall)*2
        rlm      rtall
        rlm      rtalh
        ;-----
        lcall    add
        lcall    add
        ret
;
;
;=====
;三个数由小到大排列子程序
;该子程序实现将 min,mid,max 三个 ram 中的 8 位二进制数按

```

;由小到大的顺序排列,排列后的结果放在 min,mid 和 max 中

=====

sort_sub:

 movla 02h

 movam i1

on:

 movm min,w

 subam mid,w

 btmsc status,c

 lgoto next

 movm min,w

 movam buf

 movm mid,w

 movam min

 movm buf,w

 movam mid

 lgoto next1

next:

 movm mid,w

 subam max,w

 btmsc status,c

 lgoto next1

 movm mid,w

 movam buf

 movm max,w

 movam mid

 movm buf,w

 movam max

next1:

 decmsz i1

 lgoto on

 ret

;

=====

 end

;


=====


;范例

;=====

;系统上电后 4s 内数码管显示"88",4s 后显示所测温度值 ,
;如此时传感器阻值为 11.20k(22'C 所对应的阻值),则数码
;显示温度值应为"22".在程序运行的任何时刻,若开路或
;短路传感器或比对电阻,则数码管秒闪显示"E",表示传感器
;线路出错。

应用线路图



	Title: 40 测温DEMO线路图		
	Size: B	Number:	Revision:
	Date:		Sheet of
	File:		Drawn by: